# $Id: PKGBUILD 70215 2012-05-02 12:56:49Z spupykin $
# Maintainer: Sergej Pupykin <pupykin.s+arch@gmail.com>

pkgname=aumix
pkgver=2.9.1
pkgrel=4
pkgdesc="A color text mode sound mixer with GPM support"
arch=('i686' 'x86_64')
license=('GPL')
depends=()
makedepends=('gtk2' 'gpm' 'ncurses')
optdepends=('gtk2: aumix'
	    'ncurses: aumix-nox'
	    'gpm: aumix-nox')
replaces=('aumix-gtk')
provides=('aumix-gtk')
install=aumix.install
url=(http://www.jpj.net/~trevor/aumix.html)
source=(http://www.jpj.net/~trevor/aumix/releases/aumix-$pkgver.tar.bz2
	aumix.desktop)
md5sums=('34f28ae1c94fc5298e8bb2688c4b3a20'
         'afba8b39b8dd95d8a9d74356023de14a')

build() {
  cd $srcdir/aumix-$pkgver

  msg "No X11 build"
  ./configure --prefix=/usr --mandir=/usr/share/man --without-gtk --without-gtk1 \
	--without-alsa --without-x
  make
  install -D -m0755 src/aumix $pkgdir/usr/bin/aumix-nox
  make distclean

  msg "X11+GTK2 build"
  ./configure --prefix=/usr --mandir=/usr/share/man --without-gtk1 --without-alsa \
	--without-gpm
  make
  make DESTDIR=$pkgdir install
  install -D -m0644 $srcdir/aumix.desktop $pkgdir/usr/share/applications/aumix.desktop
}
